
*                               IABC1=.5m0
*        IABC0  = .1mA->0
*                             
*    ____|\     OUT0 |\   _______|\          |\ VOUT1
*  _|_   | \/\/\_____| \_|  INN1 | \/\/\_____| \__
*  \\\ __| /\/\/ _|_ | /       __| /\/\/  |  | /   
*     |  |/      ___ |/       |  |/       |  |/    
*     | B_OTA0   _|_   200PF  |   B_OTA1  |       
*     |          \\\ C0       |           |       
*     |_______________________|___________|   
*                     |      INP1
*        _/\  /\  /\__|
*      _|_  \/  \/
*      \\\          5K
*
.OPTIONS  GMIN=1e-18       METHOD=trap   srcsteps = 1  gminsteps = 1 trtol=1 VNTOL=1m
V_Iabc0   VIABC0     0     PWL    ( 0 .1m  1m 0 )  
B_OTA0    OUT0       0     I =   -1*v(VIABC0)*tanh( v(INP1)/.052)
C0        OUT0       0     200p   IC=1
B_BUF0    INN1       0     V =    v(OUT0)
V_Iabc1   VIABC1     0     DC .5m
B_OTA1    INP1       0     I =   -1*v(VIABC1)*tanh( (v(INP1)-v(INN1)) /.052)
R1        INP1       0     5k
B_BUF1    VOUT1      0     V =    v(INP1)
Cstray    INP1       0     15p
.tran     .1u         1m    0      .1u   UIC

.control
run
set pensize = 2
plot  v(inp1) v(out0) 
.endc

.end

=====================END_OF_SPICE============================

This code works with winspice.

Perhaps the most under exploited resource available in a
bipolar process is its dynamic range. At the time the LM13600
was designed, it as possible to build the VCO shown above 



which was practically functional from 1Hz to 1MHz. 

*                               IABC1=.5m0
*        IABC0  = .1mA->0
*                             
*    ____|\     OUT0 |\   _______|\          |\ VOUT1
*  _|_  -| \/\/\_____| \_|  INN1 | \/\/\_____| \__
*  \\\ __| /\/\/ _|_ | /       __| /\/\/  |  | /   
*     |  |/      ___ |/       | +|/       |  |/    
*     | B_OTA0   _|_   200PF  |   B_OTA1  |       
*     |          \\\ C0       |           |       
*     |_______________________|___________|   
*                     |      INP1
*        _/\  /\  /\__|
*      _|_  \/  \/
*      \\\          5K
*

Part of the reason for the LM13600 version was to be able to 
exploit this six order of magnitude dynamic range. At the 
time the LM13600 was designed, a function generator that was 
handy to find in the lab was not better.

Today, the transistors are much much faster and can work
from milli Amps to femto Amps. Thats twelve order of magnitude.
And in the BiCMOS processes, input buffers can have the 
impedance of glass. It used to be an Analog joke that linear
was in the Hz to GHz business. Today it might be possible
to build a VCO that can operate at those frequency ranges
using only internal capacitors. 


